{-# OPTIONS_GHC -fplugin=Polysemy.Plugin #-}

-- This file is part of the Wire Server implementation.
--
-- Copyright (C) 2022 Wire Swiss GmbH <opensource@wire.com>
--
-- This program is free software: you can redistribute it and/or modify it under
-- the terms of the GNU Affero General Public License as published by the Free
-- Software Foundation, either version 3 of the License, or (at your option) any
-- later version.
--
-- This program is distributed in the hope that it will be useful, but WITHOUT
-- ANY WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS
-- FOR A PARTICULAR PURPOSE. See the GNU Affero General Public License for more
-- details.
--
-- You should have received a copy of the GNU Affero General Public License along
-- with this program. If not, see <https://www.gnu.org/licenses/>.

module Spar.Sem.ScimTokenStore.Mem
  ( scimTokenStoreToMem,
  )
where

import qualified Data.Map as M
import Imports
import Polysemy
import Polysemy.State
import Spar.Scim
import Spar.Sem.ScimTokenStore

scimTokenStoreToMem ::
  Sem (ScimTokenStore ': r) a ->
  Sem r (Map ScimToken ScimTokenInfo, a)
scimTokenStoreToMem = (runState mempty .) $
  reinterpret $ \case
    Insert st sti -> modify $ M.insert st sti
    Lookup st -> gets $ M.lookup st
    LookupByTeam tid -> gets $ filter ((== tid) . (.stiTeam)) . M.elems
    UpdateName tid stid name -> modify $ M.map $ \sti ->
      if (.stiTeam) sti == tid && (.stiId) sti == stid
        then sti {stiName = name}
        else sti
    Delete tid stid -> modify $ M.filter $ \sti -> not $ (.stiTeam) sti == tid && (.stiId) sti == stid
    DeleteByTeam tid -> modify $ M.filter ((/= tid) . (.stiTeam))
